1978
USDA Forest Service General Technical Report PNW-71

PLOTTING LANDSCAPE PERSPECTIVES OF CLEARCUT UNITS

Roger H. Twito

Pacific Northwest Forest and Range Experiment Station
U.S. Department of Agriculture Forest Service
Portland, Oregon

Contents

Abstract

The esthetics of clearcut units potentially visible from a distance can be improved by judicious design. Screening clearcuts from view or shaping them to characterize natural openings are current methods used for this purpose.

Perspective plots illustrating how proposed clearcut units will look from specific off-site viewing points provide an appropriate means of evaluating how well the units meet desired visual objectives. This procedure has been automated on a desk-top computer system using digitizer, plotting peripherals, and topographic maps. The explanations, listings, and logic are included in this report.

KEYWORDS: Landscaping, clearcutting systems, logging operations analysis/design, computers (desk-top).

Introduction

Highly visible clearcuts that reduce the beauty of forested landscapes have caused increased public censure over the past decade. Corresponding efforts to protect the scenic quality of timber-producing land have resulted in revised cutting practices. These avert clearcut units which would, from distant vantage points, appear as stark, artificial, and geometric openings. Instead units are designed to resemble natural clearings or, in those places where landscapes are of higher visual importance, the clearcuts are completely screened from view.

Applying these landscape management practices to proposed clearcut units necessitates special analytical techniques. Ideally these techniques must provide a fast and convenient means of examining the visual effects of several harvest plans in order to find the best arrangement.

This paper presents a technique that is successfully meeting this need. It requires a topographic map, desk-top computer, digitizer, and plotter. The procedure creates accurate perspective illustrations (plots) of cutting unit boundary lines and other key topographic features. Its simplicity and speed make it ideal for direct operation by the landscape designer. This permits immediate interaction between the designer and the plot, with rapid "back and forth" convergence towards the best design. In addition, this technique provides information which can facilitate the field location and marking of unit boundaries. The information should be useful to those land managers concerned with both the marketable and nonmarketable values of their resource, and more specifically to landscape architects and logging engineers teamed for the purpose of planning timber harvests that meet visual objectives.

Perspective Plot Program

BASE MAP

To produce a perspective plot of a proposed clearcut unit, a contour base map similar to figure 1, including the following information, is needed.

1. Proposed clearcut unit boundary
2. Critical viewing points
3. Key ridge lines
4. Starting points for digitizing units
5. Azimuth from viewing point to the first point on cutting unit
6. Tree heights

Essential information from the base map is resolved into numeric coordinates by the digitizer, and transmitted to the computer for analysis. The result, a perspective plot, is graphically displayed on the plotter. The numeric coordinates are retained in memory for automatic replotting of the initial unit from various other viewpoints.



Figure 1.--Base map.

OUTPUT

The primary outputs of this program are the perspective plots. Figure 2 represents a sample of this output. It was generated from the base map data of figure 1.



Figure 2.--Perspective plot of clearcut unit.

A special feature of PERSPECTIVE PLOT's output is that trees are only plotted at digitized points, and these points are generally digitized only at locations where contour lines cross the unit boundary. Note in figure 2 that trees appear more widely spaced along the upper part of the unit where the boundary does not intersect as many contour lines. The user of the program, however, can easily learn to relate the plot to reality. Figure 3, which merges a photograph of the same clearcut unit with figure 2, illustrates this relationship. Note that the plotter draws a line between the base of adjacent plotted trees. This line, which represents the boundary of the unit, is only a design aid.

Nevertheless, PERSPECTIVE PLOTS do provide output which, in spite of some artificiality, represent an accurate picture of clearcut units. Once the characteristics of the output are understood, the user can accurately determine a unit's shape and visibility from key viewing points.



Figure 3.--Actual clearcut unit merged with perspective plot.

DESIGN PROCESS

In its simplest form, use of PERSPECTIVE PLOT in a design process is like looking in a mirror to see if one's grooming is adequate for public display. A higher degree of complexity, however, is introduced when logging costs are considered. Therefore, the following procedure is recommended.

Draw the proposed clearcut unit boundaries on a contour map. Then, with reference to selected viewing points, produce perspective plots of the unit.

Determine from these plots whether this unit will satisfy the visual requirements. While this evaluation may seem an inherently subjective process, professional scrutiny by landscape architects referencing well-defined visual criteria normally results in consistent judgments.

If these units do not satisfy the visual standards, they must be modified until a design has been produced that does display acceptable visual results.

Once an acceptable design is achieved, the logging costs associated with harvest of the unit should be estimated. If certain boundary irregularities add significantly to the unit's logging cost, the visual advantages added by these should be illustrated by comparative plots. If these advantages are not deemed worth the extra cost, alternative designs should be examined.

Though the concept and philosophy of this design process is quite simple, its implementation is not. An almost infinite number of clearcut boundary patterns, each with their associated logging costs, represent possible solutions. Professional skill, judgment, and teamwork are required for rapid convergence upon an acceptable plan. Use of PERSPECTIVE PLOT has greatly facilitated this design process by permitting the designer to quickly examine several reasonable approaches in the effort to select the best arrangement.

FIELD LOCATION OF UNIT

After the initial unit design has been completed, the problem remains of getting the clearcut boundaries of the unit marked on the ground to correspond with the map delineation. The PERSPECTIVE PLOT program provides helpful information for performing this task. Complete survey information (fig. 4) provides data needed to run a closed traverse survey connecting the digitized boundary points.

The key elements to this survey should include the following steps:

1. Identify on aerial photographs a point close to the proposed unit boundary.

2. Resolve the map location of this point, and a field survey tie to an identifiable elevation point on the unit boundary.

3. Begin the closed unit traverse at this point, and run the survey to correspond with the horizontal distances and azimuths as shown in figure 4. (Actual ground variations normally will cause discrepancies from the slope distance output as on figure 4.)

4. Record actual slope distance and percent.

5. When the survey is closed, compare actual elevation changes with the topographic map data shown on figure 4 type printout.

6. Good judgment may indicate the PERSPECTIVE PLOT routine be rerun using actual field survey data resolved into a corrected topographic base map.

If the harvest unit is to be yarded by long-span skylines (or other terrain sensitive methods) and requires field run skyline profiles to assure the feasibility and coverage limits of the yarding system, a different approach is indicated. It then is more efficient to locate the unit boundary from the measured skyline profiles.

LIMITATIONS

THE PERSPECTIVE PLOT program is based on plane analytical geometric derivations (appendix B) and does not compensate for the earth's curvature or atmospheric refraction. At the viewing range where the program is generally used, this correction is of

Figure 4.--Survey notes of clearcut unit.

SURVEY INFORMATION

AZIMUTH FROM VIEW POINT TO 1ST POINT ON UNIT= 205 DEGREES

AZIMUTH	SLOPE DIST.	%SLOPE	HOR. DIST.	STARTING ELEV.
236	166.0	24.8	161.1	2920
222	128.1	32.9	121.7	2960
231	110.6	38.8	103.1	3000
309	303.7	-13.3	301.0	3040
283	193.2	-21.2	189.0	3000
327	103.2	-42.0	95.2	2960
17	104.1	-41.6	96.1	2920
7	125.4	33.7	118.8	2880
346	155.7	26.6	150.5	2920
5	171.6	0.0	171.6	2960
7	265.3	0.0	265.3	2960
3	316.8	0.0	316.8	2960
358	251.1	0.0	251.1	2960
15	116.0	-36.7	108.8	2960
38	77.2	-60.6	66.0	2920
40	93.5	-47.3	84.5	2880
40	93.5	-47.3	84.5	2840
46	54.7	-107.1	37.3	2800
57	62.2	-84.0	47.6	2760
60	86.7	-52.0	77.0	2720
69	81.6	-56.3	71.1	2680
75	104.1	-41.6	96.1	2640
60	86.7	-52.0	77.0	2600
40	93.5	-47.3	84.5	2560
38	77.2	-60.6	66.0	2520
46	101.5	-42.9	93.3	2480
69	81.6	-56.3	71.1	2440
60	86.7	-52.0	77.0	2400
73	57.8	-95.8	41.7	2360
60	86.7	-52.0	77.0	2320
77	67.5	-73.5	54.4	2280
75	104.1	-41.6	96.1	2240
69	81.6	-56.3	71.1	2200
54	77.2	-60.6	66.0	2160
54	77.2	-60.6	66.0	2120
46	84.7	-53.6	74.7	2080
28	71.3	-67.8	59.0	2040
54	77.2	-60.6	66.0	2000
28	71.3	-67.8	59.0	1960
35	62.2	-84.0	47.6	1920
54	77.2	-60.6	66.0	1880
80	78.3	-59.4	67.3	1840
110	131.5	-31.9	125.2	1800
158	108.2	39.8	100.5	1760
136	44.1	214.3	18.7	1800
161	119.7	35.5	112.8	1840
163	92.6	47.9	83.5	1880
159	81.6	56.3	71.1	1920
159	81.6	56.3	71.1	1960
163	57.8	95.8	41.7	2000
163	57.8	95.8	41.7	2040
170	78.3	59.4	67.3	2080
155	71.3	67.8	59.0	2120
147	62.2	84.0	47.6	2160
163	57.8	95.8	41.7	2200
163	57.8	95.8	41.7	2240
136	68.8	71.4	56.0	2280

Figure 4.--Survey notes of clearcut unit. (Continued)

155	49.7	135.5	29.5	2320
155	47.9	151.5	26.4	2360
147	103.2	42.0	95.2	2400
155	71.3	67.8	59.0	2440
136	54.7	107.1	37.3	2480
147	62.2	84.0	47.6	2520
155	71.3	67.8	59.0	2560
155	97.2	45.2	88.5	2600
187	126.0	33.5	119.5	2640
172	89.7	49.8	80.3	2680
174	113.7	37.6	106.4	2720
163	92.6	47.9	83.5	2760
163	92.6	47.9	83.5	2800
170	77.2	60.6	66.0	2840
170	78.3	59.4	67.3	2880
159	81.6	56.3	71.1	2920
176	163.4	25.3	158.4	2960
184	254.3	15.9	251.1	3000
229	163.9	-25.2	158.9	3040
241	263.4	15.4	260.3	3000
253	169.6	18.0	167.0	3040
254	318.8	-9.5	317.3	3070
258	180.6	-22.7	176.1	3040
254	143.5	-29.0	137.8	3000
279	101.5	-42.9	93.3	2960
300	165.4	0.0	165.4	2920
275	198.4	0.0	198.4	2920

SIZE OF UNIT= 109.3 ACRES

*******************SURVEY INFORMATION COMPLETE****************

minor importance. It becomes increasingly important as the viewing range increases and intervening terrain partially screens the clearcut units. For example, if a user was plotting a clearcut unit that was 10 miles away, and by plane geometry determined that he could just see a point on the unit over the top of a closer ridge 5 miles away, an error would result. The effect of the earth's curvature and atmospheric refraction would result in this point being about 29 feet below its calculated position and below the actual line of sight. This error can be quantified by the following formula:

Vertical error in feet = $0.574\ y\ (x+y)$[1/];

where,

x = distance in miles from viewpoint to intervening ridge,
y = distance in miles from intervening ridge to clearcut unit.

An error of this magnitude would result in a slight discrepancy between PERSPECTIVE PLOT's graphic representation of the unit's exposure and reality. To whatever extent this occurs, PERSPECTIVE PLOT always errs on the conservative side, portraying more unit exposure than reality. Generally units plotted at the range where this factor becomes significant appear so tiny that anything beyond cursory analysis is of questionable value.

[1/] Van Wagner, C. E. 1965. The effect of the earth's curvature in visibility mapping. USDA For. Serv., Fire Control Notes, Vol. 26, No. 4. Washington, D.C.

The PERSPECTIVE PLOT program locates and plots topographic points regardless of whether they are actually visible or not. Therefore, the screening effect of foreground terrain must be considered, as it may hide the clearcut units. Plotting the key foreground ridges makes obvious what, if any, portion of the unit will be screened. So, while the operator is spared the tedium of plotting every grid elevation point on the base map, he must use good judgment to select the topography that is essential to the plot. Past experience with this program has shown that most operators quickly relate to the plots which portray familiar landscapes and become proficient in this regard.

User's Guide

Detailed operating instructions and the program listing are given to apply to computer systems programable in the American Standard Code for Information Interchange (ASCII) BASIC language.[2/] They were developed on a Hewlett-Packard 9830A[3/] desk-top calculator/plotter/digitizer system, but can be executed on other equivalent hardware.

The minimum Hewlett-Packard calculator system needed to run this program is:

--model 9830A calculator with 7904 words of read/write memory

--additional read-only memories

plotter control

string variables

extended I/O

--plotter (9862A)

--digitizer (9864A)

--thermal page printer (9866A)

OPERATING PROCEDURE

The operating procedures outlined in this section presume a knowledge of the Hewlett-Packard Model 9830 system procedures. Only supplemental instructions pertaining to this program will be detailed.

This program can be put on a magnetic tape cassette. The first file on this tape contains special function key statements which are important to the execution of the program. Therefore, the first step in the use of this program is:

1. Type LOAD KEY 1, press EXECUTE.

This will load instructions into the special function keys. These are labeled on the special function card (fig. 5). Explanations of the correct uses and purposes of these keys are given in table 2 (page 11).

[2/] 1977. A.N.S.I. Standard x 3.4. Amer. Nat. Stan. Lists, 20 p. New York, N.Y.

[3/] The use of trade, firm, or corporation names in this publication is not an official endorsement or approval by the U.S. Department of Agriculture of any product to the exclusion of others which may be suitable.



Figure 5.--Special function card for perspective plot program.

2. Press LOAD, 2, EXECUTE.

   This loads the main program.

3. Secure contour map to digitizer surface.

   Map must contain the input data similar to figure 1 including:

   --proposed clearcut unit(s) accurately outlined

   --critical viewing point(s) marked

   --ridge lines identified (or identifiable)

4. Place plot sheet in plotter and set margins for full 10-inch by 15-inch plot.

5. Press RUN, EXECUTE.

   At this point, the program assists the user by requesting input through visual prompters in the display.

   Continue through entries in table 1 which relate the keyboard and digitizer inputs to the visual prompters on the computer display.

Table 1--*Input explanation for perspective plot program* [1/]

Visual prompter on display	Sample keyboard response	Explanation
UNIT UPHILL OF VIEW PT? (YES/NO)	YES	The plotter draws a level line of sight mark + to reserve the bulk of the plotting space above it when the units are uphill of the viewing point, or visa versa.
ENTER DESIRED FOCAL LENGTH (IN.)	18	See figure 6 for explanation.
ENTER MAP SCALE (FEET/INCH)	1320	Refers to base map on digitizer.
ENTER CONTOUR INTERVAL (FEET)	40	Inputs contour interval from the topographic base map.
ENTER TREE HEIGHT (FEET)	140	This should be varied as the average stand height of timber surrounding the unit(s) varies to ascertain the extent of screening which will occur from the viewing point(s).
DIGITIZE VIEWING POINT	None (response via digitizer)	1. Center digitizer cursor cross hairs on the map point from which you wish to construct view of the proposed units (i.e., the viewing point). 2. After digitizer emits a short beep, press $\boxed{0}$ and then $\boxed{S}$ buttons on the cursor.
ENTER ELEVATION OF VIEWING POINT	1600	From the contour map, enter the interpolated elevation of this viewing point.
DIGITIZE 1ST POINT ON CUT UNIT	None	1. With either the template suggested in figure 6 or good judgment, pick an initial point on one cutting unit centered so desired viewing field will not fall outside of plotting capability. 2. Center cursor cross hairs near this point, preferably where a contour line intersects the proposed unit boundary. 3. Press the $\boxed{S}$ button on the digitizer cursor after hearing the beep.
ENTER ELEVATION OF 1ST POINT	3520	Enter the contour line elevation of this point.
SELECT SLOPE AND PROCEED	Press appropriate function key; e.g., f_1 UPHILL f_2 LEVEL f_3 DOWNHILL f_4 FRACTIONAL	1. The key pressed must correspond with direction of slope of the unit boundary commencing at the first point and digitizing clockwise around the unit. 2. Digitizing the unit in a counter-clockwise direction will result in erroneous output of survey information (key f_8).

[1/] Table 2 explains the role and use of special function keys.

Table 2--*Explanation of role and use of special function keys*[1/]

Key	Visual prompter on display	Sample keyboard response	Explanation
f_1	UPHILL	Digitize	1. After the digitizer beeps, press the $\boxed{S}$ button when the cursor cross hairs are centered on the contour line/unit boundary intersection immediately uphill of the last point digitized. 2. Continue moving uphill, centering the cross hairs on the contour line/unit intersections and pressing the $\boxed{S}$ button after each beep. 3. The display flashes the elevation after each point is digitized. The elevation is increased by increments equal to the contour interval with each point digitized. The elevation reading provides a check for the operator.
f_2	LEVEL	Digitize	Operate in the same manner as the f_1 key. The difference is that the elevation of all points digitized is held constant.
f_3	DOWNHILL	Digitize	Operate the same way as the f_1 key. This key, however, subtracts the contour interval from the elevation of the previously digitized point.
f_4	FRACTIONAL (+ UPHILL; - DOWNHILL)	-.5, Digitize	1. This key permits plotting of additional map points interpolated between contour lines. 2. Care must be taken to continue plotting in this mode (i.e., don't change to UPHILL or DOWNHILL mode) until the elevation equals the elevation of a plotted contour line. Example: If the contour interval is 40 feet, and the operator wants to plot 10-feet-elevation increments downhill, enter -.25 and digitize four times at interpolated quarter points between the contour lines.
	When the plotting of an individual harvest unit is complete, and it is necessary to plot additional harvest units *press: STOP,* f_5		
f_5	MARK NEW HARVEST UNIT		This message will be displayed for 2 seconds. The plotter pen will lift and the next message will be displayed.
	DIGITIZE 1ST POINT ON NEXT UNIT	Digitize	Center the cursor's cross hairs on any contour line/unit boundary intersection point on the next unit, and press the $\boxed{S}$ button after the digitizer's "beep".
	ELEVATION OF 1ST PT ON NEXT UNIT =	4000	From the contour base map, enter the elevation of the point digitized.
	SELECT A SLOPE AND PROCEED	f_1 f_2 f_3 or f_4	Press the special function key that corresponds with the contour line route which will initiate the unit boundary traverse, and proceed as previously described.

See footnote at end of table.

Table 2--*Explanation of role and use of special function keys*[1/] - continued

Key	Visual prompter on display	Sample keyboard response	Explanation
	When it is necessary to change tree height		
	press: STOP, f_6		
f_6	NEW TREE HEIGHT	90	1. Program can be interrupted at any time to make changes in the tree size. 2. The program will continue operating with the same slope (up, down, or level) as before. 3. If both a new tree size and slope change are required, first change the tree size, then press STOP again and press the special function key for the desired new slope, and continue digitizing.
	When the plotting of all the harvest units is complete, and it is necessary to plot additional topographic features		
	press: STOP, f_7		
f_7	MARK TOPOG FEATURE OR RIDGE		This message will be displayed for 2 seconds. Then the plotter pen will lift and the next message will be displayed.
	D'TIZE 1ST PT ON TOPOG FEATURE	Digitize	Center the cursor's cross hairs on any contour line point that also describes a point on the topog feature which the operator wishes to delineate.
	ELEV. OF 1ST TOPOG POINT =	4280	Enter the elevation from the contour base map of the point digitized. 1. This operates similarly to special function key f_5 and may appear to be an unnecessary duplication. This separation, however, was done for the purpose of labeling groups of digitized points for greater selectivity and control on the survey notes and re-plot features of the program. 2. If survey notes for the harvest units are needed, the operator must be cautioned to digitize all harvest units before digitizing any ridges. Otherwise the survey notes will not be printed for any units digitized after the first ridge is marked. 3. It may be necessary or advantageous to plot some ridges before some of the harvest units are plotted and still obtain survey notes for all the harvest units. This can be done simply by labeling every new ridgeline or unit as a new harvest unit. Superfluous survey notes and erroneous area calculations will be printed for the ridgelines, but the necessary survey data will also be printed for all the actual harvest units.
	To obtain a listing of the survey data required for field layout of the harvest units		
	press: STOP, f_8		

See footnote at end of table.

Table 2--*Explanation of role and use of special function keys*[1/] - continued

Key	Visual prompter on display	Sample keyboard response	Explanation
f_8	AZIMUTH-V.P. TO 1ST POINT ON UNIT	16	1. From the contour base map, determine the magnetic bearing from the last viewing point to the first point on the unit. 2. A complete printout of survey notes for all harvest units marked will be printed (fig. 4).
	When all the harvest units and key topographic features within the viewing field have been plotted and a different perspective plot of these features from a different viewing point is required, put a clean sheet of paper in the plotter and *press: STOP,* f_o		
f_o	DIGITIZE VIEWING POINT	Digitize	Center the cursor's cross hairs on the new viewing, and press the $\boxed{S}$ button after the digitizer's "beep".
	ENTER ELEVATION OF VIEWING POINT	1680	Enter the elevation from the contour map of this viewing point.
	DIGITIZE 1ST POINT ON CUT UNIT	Digitize	1. This gives the operator an opportunity to re-center the plot from the new viewing point. 2. Use the template from figure 6 for picking a first point that will provide for inclusion of all the points plotted from the initial viewing point. 3. The focal length, Q8 (Prog. Statement 150) can be changed if necessary.
	ENTER ELEVATION OF 1ST POINT	3380	Enter the contour elevation of this point.
	PLOT COMPLETE		1. This message is displayed when the previously digitized points are plotted with respect to the new viewing point. 2. This process can be performed repeatedly from additional viewing points by repetition of the outlined instructions.
	REPLOT MEMORY FULL		1. If the original plot should ever accumulate 512 topographic points in memory, this message will be displayed to warn the operator that any additional plotting will prevent re-plots from memory through special function key f_o.

[1/] Special function keys can be accessed *anytime* during the program operation by pressing STOP and the appropriate special function key.

Figure 6.--Width of field for various focal lengths (numbers are focal length values).



Focal Distance Response:

1. This establishes width of viewing field attainable with the maximum 15-inch horizontal plot limit established.
2. If 18 (inches) is entered, the plot will be constructed to appear at true viewing size if the plot is held 18 inches from the viewer's eye, and correspondingly for other values entered.
3. To use the program, construct an overlay template as shown above to aid in centering the plot, and select a focal length value that will enable the entire field of digitized data to be included on one plot. Or maybe the program user will want to have the view constructed at a larger scale and use two or more plot sheets with some overlap plotted to obtain a larger scale and/or wider angle view. In any case, the template will be advantageous for correct construction of these views.
4. The half angles measured for this template are:

30-inch focal length = 14.0°
18-inch focal length = 22.6°
14-inch focal length = 28.2°
10-inch focal length = 36.9°
8-inch focal length = 43.2°

Appendix A

PROGRAM LISTINGS

```
10 REM *****PROGRAM:PERSPECTIVE PLOT
30 DIM B$[3],A[256,8]
40 M=U=0
50 DISP "UNIT UPHILL OF VIEW PT? (YES/NO)";
60 INPUT B$
70 IF B$="YES" THEN 100
80 SCALE -7.5,7.5,-7,3
90 GOTO 110
100 SCALE -7.5,7.5,-3,7
110 PLOT -0.3,0,-2
120 PLOT 0.3,0,-1
130 PLOT 0,0.3,-2
140 PLOT 0,-0.3,-1
150 DISP "ENTER DESIRED FOCAL LENGTH (IN.)";
160 INPUT Q8
170 DISP "ENTER MAP SCALE (FEET/INCH)      ";
180 INPUT H
190 DISP "ENTER CONTOUR INTERVAL (FEET)    ";
200 INPUT C
210 DISP "ENTER TREE HEIGHT (FEET)         ";
220 INPUT K
230 DISP "DIGITIZE VIEWING POINT           ";
240 WAIT 2000
250 WRITE (9,*)
260 ENTER (9,*)X1,Y1
270 DISP "ENTER ELEVATION OF VIEWING POINT";
280 INPUT Z1
290 IF M=1 THEN 310
300 N1=1
310 DISP 'DIGITIZE 1ST POINT ON CUT UNIT  ";
320 WAIT 1000
330 WRITE (9,*)
340 ENTER (9,*)X2,Y2
350 DISP "ENTER ELEVATION OF 1ST POINT     ";
360 INPUT Z2
370 A[1,1]=X2
380 A[1,2]=Y2
390 A[1,3]=Z2
400 A[1,4]=K
410 X=X1-X2
420 Y=Y1-Y2
430 DEG
440 IF X#0 THEN 460
450 X=0.01
460 A1=ATN(Y/X)
470 Z=Z2-Z1
480 D1=SQR(X*X+Y*Y)*H
490 F=Q8*(Z/D1)
500 PLOT 0,F
```

```
510 IF N1=1 THEN 2380
520 IF M=1 THEN 1640
530 DISP "SELECT A SLOPE AND PROCEED"
540 STOP
550 WRITE (9,*)
560 ENTER (9,*)X3,Y3
570 N1=N1+1
580 Z3=Z2+C1*C
590 DISP Z3
600 WAIT 250
610 IF N1>256 THEN 1070
620 A[N1,1]=X3
630 A[N1,2]=Y3
640 A[N1,3]=Z3
650 A[N1,4]=K
660 X7=A[N1,1]-X1
670 Y7=A[N1,2]-Y1
680 Z7=A[N1,3]-Z1
690 IF X7#0 THEN 710
700 X7=0.01
710 D7=SQR(X7*X7+Y7*Y7)*H
720 A7=ATN(Y7/X7)
730 A8=A1-A7
740 W=Q8*TAN(A8)
750 V8=ATN(Z7/D7)
760 G=SQR(W*W+Q8*Q8)
770 V9=ATN((Z7+K)/D7)
780 L1=G*TAN(V8)
790 L2=G*TAN(V9)
800 R=0
810 PLOT W,L1,-2
820 PLOT W,L2
830 IF (L2-L1)<0.3 THEN 1970
840 T1=0.2*(L2-L1)
850 T2=0.5*(L2-L1)
860 IPLOT -T1*0.25,-T2*0.25,0
870 IPLOT T1*0.2,T2*0.05,0
880 IPLOT -T1*0.35,-T2*0.3,0
890 IPLOT T1*0.25,T2*0.05,0
900 IPLOT -T1*0.3,-T2*0.3,0
910 IPLOT T1*0.3,T2*0.05,0
920 IPLOT -T1*0.35,-T2*0.3,0
930 IPLOT T1*0.5,T2*0.1,0
940 IPLOT T1*0.5,-T2*0.1,0
950 IPLOT -T1*0.35,T2*0.3,0
960 IPLOT T1*0.3,-T2*0.05,0
970 IPLOT -T1*0.3,T2*0.3,0
980 IPLOT T1*0.25,-T2*0.05,0
990 IPLOT -T1*0.35,T2*0.3,0
1000 IPLOT T1*0.2,-T2*0.05,0
```

```
1010 IPLOT -T1*0.25,T2*0.25,0
1020 PLOT W,L1
1030 IF R=1 THEN 2060
1040 Z2=Z3
1050 IF M=2 THEN 2350
1060 GOTO 550
1070 M1=N1-256
1080 A[M1,5]=X3
1090 A[M1,6]=Y3
1100 A[M1,7]=Z3
1110 A[M1,8]=K
1120 IF N1=512 THEN 1170
1130 X7=A[M1,5]-X1
1140 Y7=A[M1,6]-Y1
1150 Z7=A[M1,7]-Z1
1160 GOTO 690
1170 DISP "REPLOT MEMORY FULL"
1180 STOP
1190 N1=1
1200 GOTO 690
1210 DISP "UPHILL"
1220 WAIT 1000
1230 C1=1
1240 GOTO 550
1250 DISP "LEVEL SLOPE"
1260 WAIT 1000
1270 C1=0
1280 GOTO 550
1290 DISP "DOWNHILL"
1300 WAIT 1000
1310 C1=-1
1320 GOTO 550
1330 DISP "FRACTIONAL(+UPHILL; -DOWNHILL)";
1340 INPUT C1
1350 GOTO 550
1360 DISP "NEW TREE HEIGHT";
1370 INPUT K
1380 GOTO 550
1390 DISP "MARK TOPOG FEATURE OR RIDGE"
1400 WAIT 2000
1410 N1=N1+1
1420 IF N1>256 THEN 1450
1430 A[N1,1]=A[N1,2]=A[N1,3]=A[N1,4]=-1
1440 GOTO 1470
1450 M1=N1-256
1460 A[M1,5]=A[M1,6]=A[M1,7]=A[M1,8]=-1
1470 PEN
1480 DISP "D'TIZE 1ST PT. ON TOPOG FEATURE";
1490 WAIT 2000
1500 WRITE (9,*)
```

```
1510 ENTER (9,*)X3,Y3
1520 DISP "ELEV. OF 1ST TOPOG POINT=";
1530 INPUT Z3
1540 N1=N1+1
1550 M=2
1560 GOTO 610
1570 M=1
1580 PEN
1590 IF U=0 THEN 1620
1600 N1=N1-1
1610 GOTO 1630
1620 N2=N1
1630 GOTO 230
1640 FOR N1=2 TO N2
1650 IF N1>256 THEN 1710
1660 X7=A[N1,1]-X1
1670 Y7=A[N1,2]-Y1
1680 Z7=A[N1,3]-Z1
1690 K=A[N1,4]
1700 GOTO 1790
1710 M1=N1-256
1720 X7=A[M1,5]-X1
1730 Y7=A[M1,6]-Y1
1740 Z7=A[M1,7]-Z1
1750 K=A[M1,8]
1760 IF A[M1,7]=0 THEN 2110
1770 IF A[M1,7]=-1 THEN 2110
1780 GOTO 1810
1790 IF A[N1,3]=0 THEN 2110
1800 IF A[N1,3]=-1 THEN 2110
1810 IF X7#0 THEN 1830
1820 X7=0.01
1830 D7=SQR(X7*X7+Y7*Y7)*H
1840 A7=ATN(Y7/X7)
1850 A8=A1-A7
1860 W=Q8*TAN(A8)
1870 V8=ATN(Z7/D7)
1880 G=SQR(W*W+Q8*Q8)
1890 V9=ATN((Z7+K)/D7)
1900 L1=G*TAN(V8)
1910 L2=G*TAN(V9)
1920 IF N1=2 THEN 2130
1930 PLOT W,L1,-2
1940 PLOT W,L2
1950 R=1
1960 IF (L2-L1) >= 0.3 THEN 840
1970 T1=0.2*(L2-L1)
1980 T2=0.5*(L2-L1)
1990 IPLOT -T1/4,-T1/4,0
2000 IPLOT -T1/4,-(T2-(T1/4)),0
```

```
2010 IPLOT T1,0,0
2020 IPLOT -T1/4,(T2-(T1/4)),0
2030 IPLOT -T1/4,T1/4,0
2040 PLOT W,L1
2050 IF R=0 THEN 1040
2060 NEXT N1
2070 PEN
2080 U=1
2090 DISP "PLOT COMPLETE"
2100 STOP
2110 PEN
2120 GOTO 2060
2130 PEN
2140 GOTO 1930
2150 DISP "PLOT COMPLETE"
2160 STOP
2170 DISP "MARK NEW HARVEST UNIT"
2180 WAIT 2000
2190 N1=N1+1
2200 IF N1>256 THEN 2230
2210 A[N1,1]=A[N1,2]=A[N1,3]=A[N1,4]=0
2220 GOTO 2250
2230 M1=N1-256
2240 A[M1,5]=A[M1,6]=A[M1,7]=A[M1,8]=0
2250 PEN
2260 DISP "DIGITIZE 1ST POINT ON NEXT UNIT ";
2270 WAIT 2000
2280 WRITE (9,*)
2290 ENTER (9,*)X3,Y3
2300 DISP "ELEV. OF 1ST PT ON NEXT UNIT = ";
2310 INPUT Z3
2320 N1=N1+1
2330 M=2
2340 GOTO 610
2350 M=0
2360 DISP "SELECT SLOPE AND PROCEED"
2370 STOP
2380 N1=N1+1
2390 A[N1,1]=A[1,1]
2400 A[N1,2]=A[1,2]
2410 A[N1,3]=A[1,3]
2420 A[N1,4]=A[1,4]
2430 GOTO 520
2440 PRINT TAB25"SURVEY INFORMATION"
2450 PRINT
2460 DISP "AZIMUTH-V.P. TO 1ST PT ON UNIT";
2470 INPUT B1
2480 WRITE (15,2490)B1
2490 FORMAT "AZIMUTH FROM VIEW POINT TO 1ST POINT
                             ON UNIT=",F5.0,"  DEGREES",/
2500 PRINT "  AZIMUTH     SLOPE DIST.     %SLOPE
                            HOR. DIST.     STARTING ELEV. "
```

```
2510 PRINT
2520 N2=N1
2530 J1=B=0
2540 FOR N1=(2+B) TO N2-1
2550 IF N1=256 THEN 3060
2560 IF N1>256 THEN 3130
2570 P=A[N1,1]-A[N1+1,1]
2580 P1=A[N1,2]-A[N1+1,2]
2590 P3=A[N1,3]-A[N1+1,3]
2600 J=(A[N1,2]*A[N1+1,1])-(A[N1,1]*A[N1+1,2])
2610 IF A[N1+1,3]=0 THEN 2900
2620 IF A[N1+1,3]=-1 THEN 2900
2630 E1=A[N1,3]
2640 L=SQR(P↑2+P1↑2)*H
2650 P4=(P3/L)*-100
2660 P5=SQR(L↑2+P3↑2)
2670 IF P#0 THEN 2690
2680 P=0.01
2690 P6=ATN(P1/P)
2700 IF P<0 THEN 2730
2710 P7=270-P6
2720 GOTO 2740
2730 P7=90-P6
2740 IF X>0 THEN 2770
2750 B2=B1-(90-A1)
2760 GOTO 2780
2770 B2=B1-(270-A1)
2780 P8=P7+B2
2790 IF P8<0 THEN 2830
2800 IF P8>360 THEN 2850
2810 P9=P8
2820 GOTO 2860
2830 P9=P8+360
2840 GOTO 2860
2850 P9=P8-360
2860 J1=J+J1
2870 WRITE (15,2880)P9,P5,P4,L,E1
2880 FORMAT F6.0,F15.1,F12.1,F11.1,F15.0
2890 NEXT N1
2900 J2=J1*H↑2/87120
2910 PRINT
2920 WRITE (15,2930)J2
2930 FORMAT "       SIZE OF UNIT=",F8.1," ACRES",/
2940 J1=0
2950 B=N1
2960 IF N1=N2 THEN 3020
2970 IF N1 >= 256 THEN 3000
2980 IF A[N1+1,3]=-1 THEN 3020
2990 GOTO 2540
3000 IF A[N1-255,7]=-1 THEN 3020
```

```
3010 GOTO 2540
3020 PRINT "******SURVEY INFORMATION COMPLETE******"
3030 PRINT
3040 PRINT
3050 STOP
3060 P=A[N1,1]-A[N1-255,5]
3070 P1=A[N1,2]-A[N1-255,6]
3080 P3=A[N1,3]-A[N1-255,7]
3090 J=(A[N1,2]*A[N1-255,5])-(A[N1,1]*A[N1-255,6])
3100 IF A[N1-255,5]=0 AND A[N1-255,6]=0 THEN 2900
3110 IF A[N1-255,7]=-1 THEN 2900
3120 GOTO 2630
3130 M1=N1-256
3140 P=A[M1,5]-A[M1+1,5]
3150 P1=A[M1,6]-A[M1+1,6]
3160 P3=A[M1,7]-A[M1+1,7]
3170 J=(A[M1,6]*A[M1+1,5])-(A[M1,5]*A[M1+1,6])
3180 IF A[M1+1,5]=0 AND A[M1+1,6]=0 THEN 2900
3190 IF A[M1+1,7]=-1 THEN 2900
3200 E1=A[M1,7]
3210 GOTO 2640
3220 STOP
```

SF KEY 0

CONT 1570

SF KEY 1

CONT 1210

SF KEY 2

CONT 1250

SF KEY 3

CONT 1290

SF KEY 4

CONT 1330

SF KEY 5

CONT 2170

SF KEY 6

CONT 1360

SF KEY 7

CONT 1390

SF KEY 8

CONT 2440

Appendix B

PROGRAM ALGORITHM

The key steps in the program are outlined to provide understanding of the mathematical algorithm. This is included to help those that may want to apply this technique using small computers that are not programable in ASCII BASIC.

This algorithm can be briefly described as relating solved horizontal and vertical angles from contour map input data to a plotted diagram where the viewing point is a fixed distance from the diagram. This distance, though variable, is set at 18 inches for this example.

1. Tape base map to digitizer.
2. Program the plotter to mark a point near the center of the sheet for the horizontal line of sight (a + mark is used for this in the present program). With a limited plot size, the mark should be nearer to the bottom of the plot if the plotted terrain is at a higher elevation than the viewing point, and visa versa.
3. Enter and store: H (map scale in feet/inch).
4. Enter and store: K (tree height in feet).
5. Enter and store: Digitized coordinates of the viewing point X1 and Y1 (see fig. 7).
6. Enter and store: Elevation in feet of the viewing point Z1.
7. Mark an intial point where a contour line crosses the unit boundary.
8. Enter and store: Digitized coordinates of the initial point X2 and Y2 (see fig. 7).
9. Enter and store: Elevation of initial point Z2.

(COMPUTER NOW CALCULATES PLOT COORDINATES FOR THIS INITIAL POINT BY THE FOLLOWING PROCESS.)

I. *Calculate horizontal distance from viewpoint to initial point.*

   a. $X = X2 - X1,$

   b. $Y = Y2 - Y1,$

   c. $D1 = (X^2 + Y^2)^{1/2} H,$

   $D1$ = *true distance in feet from viewpoint to initial point.*

II. *Calculate azimuth angle from viewpoint to initial point.*

   a. $A1 = \text{arc tan } Y/X.$

III. a. $Z = Z2 - Z1,$

   b. $F = 18\ Z/D1,$

   F = *vertical plot in inches from horizontal reference to the ground line at the initial point, and*

   $F1 = 18\ (Z + K)/D1,$

   $F1$ = *vertical plot in inches from horizontal reference to tree top at initial point.*

   The F and F1 values can be derived from proportional triangles, i.e.,



Figure 7.--Base map geometry.



true horizontal distance (viewpoint) to initial point)

and



viewer distance from plot diagram and likewise for F1 value.

IV. To plot these calculated values for the initial point:

a. Origin must be set at horizontal line of site mark.



b. The following plot commands are needed:

(LIFT PEN)

PLOT 0 (x coord.), F (y coord.),

(which plots to base of tree at initial point),

PLOT 0, F1,

(plots to top of tree at initial point),

PLOT 0, F,

(returns plotter to base of tree).

---

10. Input additional data needed to generate plot for the point adjacent to the initial point (see fig. 7).

11. Enter and store: Digitized coordinates of the next point X3 and Y3.

12. Enter and store: Elevation of next point Z3.

(COMPUTER NOW CALCULATES PLOT COORDINATES FOR THE NEXT POINT BY THE FOLLOWING PROCESS.)

I. Calculate distance from viewing point to second point:

a. $X7 = X3 - X1$,

b. $Y7 = Y3 - Y1$,

c. $D7 = (X7^2 + Y7^2)^{1/2}H$,

$D7$ = *true distance in feet from viewing point to second point.*

II. Calculate azimuth angle from viewpoint to second point:

a. $A7 = \text{arc tan } Y7/X7$.

III. Calculate azimuth angle difference between initial point and second point:

a. $A8 = A1 - A7$.

IV. Calculate "x" value to be plotted:

a. $W = 18 \tan A8$,

W = *"x" value to be plotted.*

(based on triangular relationship)



V. *Calculate "y" values to be plotted:*

a. *V8 = arc tan Z7/D7,*

and

V9 = arc tan (Z7 + K)/D7.

b. $G = (W^2 + 18^2)^{1/2}$,

G = horizontal distance from eye to "x" projection of second plotted point.

c. *L1 = G tan V8,*

L1 = "y" coordinate to plot for base of tree, and

L2 = G tan V9,

L2 = "y" coordinate to plot for top of tree.

VI. *To plot the values now calculated for the second point, the following plot commands are needed:*

(Pen down to draw line from initial point to second point),

PLOT W (x coord.), L1 (y coord.)

(which plots to base of tree at second point),

PLOT W, L2

(plots to top of tree at second point),

PLOT W, L1

(returns plotter to base of tree).

---

13. Repeat steps 10 through 12 for each additional point required.

If several clearcut units and ridgetops are plotted from one viewing point, it is desirable to lift the plotter's pen through program control or manually as it plots from one unit to the next.

Most desk-top systems can be programed to automatically calculate the elevation at each plotted point except for the initial one. This time saving modification requires entering and storing the contour line interval in feet and arranging a technique to signal the program for plotting downhill (decrementing elevation), uphill (incrementing elevation), or level (holding elevation constant).

GPO 986-501

The mission of the PACIFIC NORTHWEST FOREST AND RANGE EXPERIMENT STATION is to provide the knowledge, technology, and alternatives for present and future protection, management, and use of forest, range, and related environments.

Within this overall mission, the Station conducts and stimulates research to facilitate and to accelerate progress toward the following goals:

1. Providing safe and efficient technology for inventory, protection, and use of resources.
2. Developing and evaluating alternative methods and levels of resource management.
3. Achieving optimum sustained resource productivity consistent with maintaining a high quality forest environment.

The area of research encompasses Oregon, Washington, Alaska, and, in some cases, California, Hawaii, the Western States, and the Nation. Results of the research are made available promptly. Project headquarters are at:

Fairbanks, Alaska	Portland, Oregon
Juneau, Alaska	Olympia, Washington
Bend, Oregon	Seattle, Washington
Corvallis, Oregon	Wenatchee, Washington
La Grande, Oregon	

Mailing address: Pacific Northwest Forest and Range Experiment Station
P.O. Box 3141
Portland, Oregon 97208